# ===========================================================================
#             http://autoconf-archive.cryp.to/ax_boost_regex.html
# ===========================================================================
#
# SYNOPSIS
#
#   AX_BOOST_REGEX
#
# DESCRIPTION
#
#   Test for Regex library from the Boost C++ libraries. The macro requires
#   a preceding call to AX_BOOST_BASE. Further documentation is available at
#   <http://randspringer.de/boost/index.html>.
#
#   This macro calls:
#
#     AC_SUBST(BOOST_REGEX_LIB)
#
#   And sets:
#
#     HAVE_BOOST_REGEX
#
# LAST MODIFICATION
#
#   2008-04-12
#
# COPYLEFT
#
#   Copyright (c) 2008 Thomas Porschberg <thomas@randspringer.de>
#   Copyright (c) 2008 Michael Tindal
#
#   Copying and distribution of this file, with or without modification, are
#   permitted in any medium without royalty provided the copyright notice
#   and this notice are preserved.

AC_DEFUN([AX_BOOST_REGEX],
[
    AC_ARG_WITH([boost-regex],
    AS_HELP_STRING([--with-boost-regex=special-lib],
                   [specify a certain version of the Boost regex library for the linker e.g. --with-boost-regex=boost_regex-gcc-mt-d-1_33_1 ]),
        [
        if test "$withval" = "no"; then
            want_boost="no"
        elif test "$withval" = "yes"; then
            want_boost="yes"
            ax_boost_user_regex_lib=""
        else
            want_boost="yes"
            ax_boost_user_regex_lib="$withval"
        fi
        ],
        [want_boost="yes"]
    )

    if test "x$want_boost" != "xyes"; then
        AC_MSG_ERROR(The Boost regex library is required for building MKVToolNix)
    fi
    AC_REQUIRE([AC_PROG_CC])
    CPPFLAGS_SAVED="$CPPFLAGS"
    CPPFLAGS="$CPPFLAGS $BOOST_CPPFLAGS"
    export CPPFLAGS

    LDFLAGS_SAVED="$LDFLAGS"
    LDFLAGS="$LDFLAGS $BOOST_LDFLAGS"
    export LDFLAGS

    AC_CACHE_CHECK(whether the Boost::Regex library is available,
                   ax_cv_boost_regex,
    [AC_LANG_PUSH([C++])
         AC_COMPILE_IFELSE([AC_LANG_PROGRAM([[@%:@include <boost/regex.hpp>]],
                                            [[boost::regex r(); return 0;]])],
               ax_cv_boost_regex=yes, ax_cv_boost_regex=no)
     AC_LANG_POP([C++])
    ])
    if test "x$ax_cv_boost_regex" = "xyes"; then
        AC_DEFINE(HAVE_BOOST_REGEX,,[define if the Boost::Regex library is available])
        BOOSTLIBDIR=`echo $BOOST_LDFLAGS | sed -e 's/@<:@^\/@:>@*//'`
        if test "x$ax_boost_user_regex_lib" = "x"; then
            for libextension in `ls $BOOSTLIBDIR/libboost_regex*.{so,a,dylib}* 2>/dev/null | sed 's,.*/,,' | sed -e 's;^lib\(boost_regex.*\)\.so.*$;\1;' -e 's;^lib\(boost_regex.*\)\.a*$;\1;' -e 's;^lib\(boost_regex.*\)\.dylib.*$;\1;'` ; do
                 ax_lib=${libextension}
                AC_CHECK_LIB($ax_lib, exit,
                             [BOOST_REGEX_LIB="-l$ax_lib"; AC_SUBST(BOOST_REGEX_LIB) link_regex="yes"; break],
                             [link_regex="no"])
            done
            if test "x$link_regex" != "xyes"; then
            for libextension in `ls $BOOSTLIBDIR/boost_regex*.{dll,a}* 2>/dev/null | sed 's,.*/,,' | sed -e 's;^\(boost_regex.*\)\.dll.*$;\1;' -e 's;^\(boost_regex.*\)\.a*$;\1;'` ; do
                 ax_lib=${libextension}
                AC_CHECK_LIB($ax_lib, exit,
                             [BOOST_REGEX_LIB="-l$ax_lib"; AC_SUBST(BOOST_REGEX_LIB) link_regex="yes"; break],
                             [link_regex="no"])
            done
            fi

        else
           for ax_lib in $ax_boost_user_regex_lib boost_regex-$ax_boost_user_regex_lib; do
                  AC_CHECK_LIB($ax_lib, main,
                               [BOOST_REGEX_LIB="-l$ax_lib"; AC_SUBST(BOOST_REGEX_LIB) link_regex="yes"; break],
                               [link_regex="no"])
           done
        fi
        if test "x$link_regex" != "xyes"; then
            if test "x$ax_lib" = "x" ; then
                ax_lib="the Boost regex library"
            fi
            AC_MSG_ERROR(Could not link against $ax_lib !)
        fi
    fi

    CPPFLAGS="$CPPFLAGS_SAVED"
    LDFLAGS="$LDFLAGS_SAVED"
])
